
1. [bookmark: _Toc116312587]Overview
This document lists all the changes needed for baseline moving based on iwd-TTCN3-B2020-09_D22wk37.
Contact:	Olivier Genoud
	olivier.genoud@etsi.org
1. [bookmark: _Toc116312588]Baseline Moving – Required Changes
1. [bookmark: _Toc116312589]SIP/SDP types
No changes.

1. [bookmark: _Toc116312590]XSD types
1.1.1 [bookmark: _Toc116312591]XSD Modules
	IANA-resource-lists.xsd
	
	RFC 4826 clause 3.2 (no changes)

	RFC4354-poc-settings.xsd
	
	RFC 4354 clause 6.1 (no changes)

	RFC4661-SimpleFilter.xsd
	
	RFC 4661 clause 7 (no changes)

	RFC5261-patch-ops.xsd
	
	RFC 5261 clause 8 (no changes)

	RFC5874-xcap-diff.xsd
	
	RFC 5874 clause 4 (no changes)

	poc_listService-v1_0.xsd
	
	OMA - poc_listService-v1_0: "List Service" (no changes)

	xdm_extensions-v1_0.xsd
	
	OMA - xdm_extensions-v1_0: "XDM - XDM2 - Extensions" (no changes)

	xdm_rsrclst_uriusage-v1_0.xsd
	
	OMA - xdm_commonPolicy-V1_0: "XDM - Common Policy" (no changes)

	xenc-schema.xsd
	
	http://www.w3.org/2001/04/xmlenc#' (no changes)

	xmldsig-core-schema.xsd
	
	http://www.w3.org/2000/09/xmldsig# (no changes)

	TS24281_mcvideoinfo.xsd
	V17.8.0
	Updated; no impact on TTCN

	TS24281_mcvideolocationinfo.xsd
	V17.8.0
	Updated; no impact on TTCN

	TS24281_mcvideoPresInfo.xsd
	V17.8.0
	Updated; no impact on TTCN

	TS24282_mcdatainfo.xsd
	V17.8.0
	Updated; no impact on TTCN

	TS24282_mcdatalocationinfo.xsd
	V17.8.0
	Updated; no impact on TTCN

	TS24282_mcdataPresInfo.xsd
	V17.8.0
	Updated; no impact on TTCN

	TS24379_mcpttaff.xsd
	V17.8.0
	No changes

	TS24379_mcpttinfo.xsd
	V17.8.0
	Updated; no impact on TTCN

	TS24379_mcpttlocationinfo.xsd
	V17.8.0
	Updated; no impact on TTCN

	TS24379_mcpttPresInfo.xsd
	V17.8.0
	Updated; no impact on TTCN

	TS24379_mcpttPresInfoFA.xsd
	V17.8.0
	No changes

	TS24379_mcpttsigneddoc.xsd
	V17.8.0
	No changes

	TS24379_poc-settings.xsd
	V17.8.0
	No changes

	TS24481-mcptt-group.xsd
	V17.5.0
	Updated; no impact on TTCN

	TS24481-mcpttGKTP.xsd
	V17.5.0
	No changes

	TS24481-mcpttGMOP.xsd
	V17.5.0
	No changes

	TS24484-mcdata-service-config.xsd
	V17.7.0
	Updated; no impact on TTCN

	TS24484-mcdata-ue-config.xsd
	V17.7.0
	No changes

	TS24484-mcdata-user-profile.xsd
	V17.7.0
	Updated; no impact on TTCN

	TS24484-mcptt-service-config.xsd
	V17.7.0
	Updated; no impact on TTCN

	TS24484-mcptt-ue-config.xsd
	V17.7.0
	No changes

	TS24484-mcptt-user-profile.xsd
	V17.7.0
	Updated; no impact on TTCN

	TS24484-mcvideo-service-config.xsd
	V17.7.0
	Updated; requires TTCN changes as shown in 2.3.2

	TS24484-mcvideo-ue-config.xsd
	V17.7.0
	No changes

	TS24484-mcvideo-user-profile.xsd
	V17.7.0
	Updated; no impact on TTCN

	TS24484-ue-init-config.xsd
	V17.7.0
	Updated; requires TTCN changes as shown in 2.3.1

	TS33180-mcsecKMSInterface.xsd
	V17.7.0
	No changes

	TS33180-mcsecKMSKRR.xsd
	V17.7.0
	No changes


1.1.2 [bookmark: _Toc116312592]TTCN Changes
1.1.2.1 [bookmark: _Toc116312593]cs_MCPTT_UE_InitialConfiguration_OnNetwork
	Reason for change
	New optional fields in On-networkType

	Summary of change
	New optional fields added to cs_MCPTT_UE_InitialConfiguration_OnNetwork

	TTCN module
	MCX_XML_Configuration_Initial



	  template (value) MCPTT_UE_InitialConfiguration_OnNetwork_Type cs_MCPTT_UE_InitialConfiguration_OnNetwork(charstring p_PLMN) :=
  { /* @status    APPROVED (MCX) */
    index := omit,
    attr := omit,
    timers := {
      t100 := 2,
      t101 := 2,
      t103 := 5,
      t104 := 2,
      t132 := 3,
      anyExt := omit,
      elem_list := {}
    },
    hPLMN := {                          // @sic R5s201409 BASELINE MOVING 2020 hPLM -> hPLMN sic@
      pLMN := p_PLMN,
      service := cs_MCPTT_UE_InitialConfiguration_Service,
      vPLMN_list := {},                  // @sic R5s201409 BASELINE MOVING 2020 vPLM -> vPLMN sic@  @sic R5-217985: no VPLMN sic@
      anyExt := omit,
      elem_list := {}
    },
    app_Server_Info := {
      idms_auth_endpoint  := f_MCX_AbsoluteServerURI(-, px_MCX_IdMS_auth_IPAddress,  px_MCX_IdMS_auth_Port,  tsc_MCX_IdMS_auth_UriPath),   // @sic R5-200385 sic@
      idms_token_endpoint := f_MCX_AbsoluteServerURI(-, px_MCX_IdMS_token_IPAddress, px_MCX_IdMS_token_Port, tsc_MCX_IdMS_token_UriPath),  // @sic R5-200385 sic@
      http_proxy := f_MCX_AbsoluteServerURI(-, px_MCX_HTTP_Proxy_IPAddress, px_MCX_HTTP_Proxy_Port),
      gms := tsc_MCX_GMS_Hostname,      // @sic R5-200385 sic@
      cms := tsc_MCX_CMS_Hostname,      // @sic R5-200385 sic@
      kms := tsc_MCX_KMS_Hostname,      // @sic R5-200385 sic@
      tls_tunnel_auth_method := {
        mutual_authentication := false,
        x509 := omit,
        key := omit,
        anyExt := omit,
        elem_list := {}
      },
      anyExt := omit,
      elem_list := {}
    },
    gMS_URI := tsc_MCX_GMSURI,
    group_creation_XUI := px_MCX_GroupCreationXUI,
    gMS_XCAP_root_URI := tsc_MCX_GMSXCAPRootURI,
    cMS_XCAP_root_URI := tsc_MCX_CMSXCAPRootURI,
    integrity_protection_enabled := tsc_MCX_XML_IntegrityProtection,
    confidentiality_protection_enabled := tsc_MCX_XML_ConfidentialityProtection,
    anyExt := omit,
    elem_list := {}
  };



1.1.2.2 [bookmark: _Toc116312594]cs_MCVideo_ServiceConfiguration_OnNetwork 
	Reason for change
	In Rel-17 new mandatory fields for configuration of the resource priorities have been added to the on-networkType in the MCVideo service configuration.

	Summary of change
	Addition of resource priorities to cs_MCVideo_ServiceConfiguration_OnNetwork.
NOTE: A prose CR will be raised at TSGR5#97 to add the MCVideo resource priorities to 36.579-1 Table 5.5.8.8-1

	TTCN modules
	MCX_XML_Configuration_MCVideo, MCX_CommonDefs



	  const charstring tsc_MCVideo_EmergencyResourcePriorityNamespace := "mcpttp";
  const charstring tsc_MCVideo_EmergencyResourcePriorityValue := "7";
  const charstring tsc_MCVideo_ImminentPerilResourcePriorityNamespace := "mcpttp";
  const charstring tsc_MCVideo_ImminentPerilResourcePriorityValue := "4";
  const charstring tsc_MCVideo_NormalResourcePriorityNamespace := "mcpttp";
  const charstring tsc_MCVideo_NormalResourcePriorityValue := "0";

	  template (value) urn_3gpp_ns_mcvideoServiceConfig_1_0.Resource_priorityType cs_MCVideo_ResourcePriority(charstring p_ResourcePriorityNamespace,
                                                                                                          charstring p_ResourcePriorityValue) :=
  {
    attr := omit,
    resource_priority_namespace := p_ResourcePriorityNamespace,
    resource_priority_priority := p_ResourcePriorityValue,
    anyExt := omit,
    elem_list := {}
  };

	  template (value) MCVideo_ServiceConfiguration_OnNetwork_Type cs_MCVideo_ServiceConfiguration_OnNetwork :=
  {
    attr := omit,
    signalling_protection := cs_MCVideo_ServiceConfiguration_Signalling_protection,
    protection_between_mcvideo_servers := cs_MCVideo_ServiceConfiguration_Server_protection,
    emergency_resource_priority := cs_MCVideo_ResourcePriority(tsc_MCVideo_EmergencyResourcePriorityNamespace, tsc_MCVideo_EmergencyResourcePriorityValue),
    imminent_peril_resource_priority := cs_MCVideo_ResourcePriority(tsc_MCVideo_ImminentPerilResourcePriorityNamespace, tsc_MCVideo_ImminentPerilResourcePriorityValue),
    normal_resource_priority := cs_MCVideo_ResourcePriority(tsc_MCVideo_NormalResourcePriorityNamespace, tsc_MCVideo_NormalResourcePriorityValue),
    anyExt := omit,
    elem_list := {}
  };



1. [bookmark: _Toc116312595]TTCN type definitions
1.2.1 [bookmark: _Toc116312596]MIKEY Messages (RFC 6509 and RFC 6043)
No changes in MIKEY_TypeDefs.
1.2.2 [bookmark: _Toc116312597]MIKEY general extension payload (33.180 V17.7.0 E.6)
No changes in MIKEY_MCX_Extension_TypeDefs.
1.2.3 [bookmark: _Toc116312598]MCPTT Floor Control and Preestablished session control (24.380 V17.7.0 clause 8)
1.2.3.1 [bookmark: _Toc116312599]FloorControl_ListOfSSRCs
	Reason for change
	Change of 24.380 Table 8.2.3.18-1

	Summary of change
	Update of TTCN Definitions
NOTE 1: The change of FloorControl_ListOfSSRCs is non-backward compatible as 'lgth' now is the number of octets (i.e. four times the number of SSRCs); this change has not been done for Rel-17 only but also for previous releases (e.g.Rel-15)
NOTE 2: The 'lgth' field is not specified for any IE of 36.579-1 clause 5.5.6 but needs to be considered by the codec.

	TTCN module
	MCPTT_MediaControl_TypeDefs, MCPTT_MediaControl_Templates



	  type record FloorControl_ListOfSSRCs {                        // Rel-15
    O1_Type id,
    O1_Type lgth,
    O1_Type noOfSSRCs,
    O2O1_Type spare,
    RTP_SSRC_List list
  };

	  template (value) FloorControl_ListOfSSRCs cs_FloorControl_ListOfSSRCs(RTP_SSRC_List p_SSRC_List) :=
  {
    id := '10'O,
    lgth := int2oct(lengthof(p_SSRC_List) * 4, 1),
    noOfSSRCs := int2oct(lengthof(p_SSRC_List), 1),
    spare := '0000'O,
    list := p_SSRC_List
  };



1.2.3.2 [bookmark: _Toc116312600]FloorControl_Ack
	Reason for change
	Changes of 24.380 Table 8.2.13-1 for Rel-16/17 

	Summary of change
	Update of TTCN Definitions
NOTE: Change of cs_FloorCtrlMsg_Ack and cr_FloorCtrlMsg_Ack is subject to RAN5 approval.

	TTCN module
	MCPTT_MediaControl_TypeDefs, MCPTT_MediaControl_Templates



	  type record FloorControl_Ack {
    FloorControl_CommonHdr commonHdr,
    FloorControl_Source source,
    FloorControl_MsgType msgType,
    FloorControl_TrackInfo trackInfo optional,
    FloorControl_Location location optional
  };

	  template (value) FloorControlMsg_Type cs_FloorCtrlMsg_Ack(O1_Type p_MsgType) :=
  {
    ackMessage := {
      commonHdr := cs_FloorControl_CommonHdr('01010'B, tsc_MCX_SSRC_NW),         // ACK subtype 24.380 clause 8.2.2.1
      source := cs_FloorControl_Source('0002'O),
      msgType := cs_FloorControl_MsgType(p_MsgType),
      trackInfo := omit,
      location := omit
    }
  };

	  template (present) FloorControlMsg_Type cr_FloorCtrlMsg_Ack(template (present) RTP_SSRC p_SSRC := ?,
                                                              O1_Type p_MsgType) :=
  {
    ackMessage := {
      commonHdr := cr_FloorControl_CommonHdr('01010'B, p_SSRC),                 // ACK subtype 24.380 clause 8.2.2.1
      source := cs_FloorControl_Source('0000'O),
      msgType := cr_FloorControl_MsgType(p_MsgType),
      trackInfo := omit,
      location := cr_FloorControl_Location(crs_FloorControl_LocationValue_NotProvided) ifpresent
    }
  };



1.2.3.3 [bookmark: _Toc116312601]PreestablishedSessionControl_Disconnect
	Reason for change
	Changes of 24.380 Table 8.3.5-1 for Rel-17 

	Summary of change
	Update of TTCN Definitions
NOTE: 24.380 does not specify when the Reason cause shall be provided to the client and what the client shall do with it.

	TTCN module
	MCPTT_MediaControl_TypeDefs, MCPTT_MediaControl_Templates



	  type record PreestablishedSessionControl_ReasonCause {
    O1_Type id,
    O1_Type lgth,
    O2_Type val
  };

	  type record PreestablishedSessionControl_Disconnect {
    PreestablishedSessionControl_CommonHdr              commonHdr,
    PreestablishedSessionControl_SessionIdentity        sessionIdentity optional,
    PreestablishedSessionControl_ReasonCause            reasonCause optional
  };

	  template (value) PreestablishedSessionControlMsg_Type cs_PreestablishedSessionControl_Disconnect(template (value) PreestablishedSessionControl_SessionIdentity p_SessionIdentity,
                                                                                                   MCX_MediaControl_AckRequest_Type p_AckRequest) :=
  {
    disconnectMessage := {
      commonHdr := cs_PreestablishedSessionControl_CommonHdr(f_MCX_MediaControlSubtype_TX('00001'B, p_AckRequest), tsc_MCX_SSRC_NW_P),
      sessionIdentity := p_SessionIdentity,
      reasonCause := omit
    }
  };



1.2.3.4 [bookmark: _Toc116312602]New Floor Control Messages
New Rel-17 type definitions are added to MCPTT_MediaControl_TypeDefs:
· FloorControl_QueuedFloorRequestsPurpose
(24.380 clause 8.2.3.23)
· FloorControl_ListOfQueuedUsers
(24.380 clause 8.2.3.24)
· FloorControl_QueuedFloorRequestsResult
(24.380 clause 8.2.3.25)
· FloorControl_MediaFlowControlIndicator
(24.380 clause 8.2.3.26)
· FloorControl_QueuedFloorRequests
(24.380 clause 8.2.15)
· FloorControl_
(24.380 clause 8.2.16)
1.2.4 [bookmark: _Toc116312603]MCVideo media plane control (24.581 V.17.2.0 clause 9)
1.2.4.1 [bookmark: _Toc116312604]New Fields for Functional Alias and Reception Mode
	Reason for change
	New clauses 9.2.3.21 and 9.2.3.22 in 24.581 for Rel-17

	Summary of change
	New types TransmissionControl_FunctionalAlias, TransmissionControl_ReceptionMode

	TTCN module
	MCVideo_TransmissionControl_TypeDefs



	  type record TransmissionControl_FunctionalAlias {
    O1_Type id,
    O1_Type lgth,
    octetstring val,
    octetstring padding
  };

	  type record TransmissionControl_ReceptionMode {
    O1_Type id,
    O1_Type lgth,
    O2_Type val
  };



1.2.4.2 [bookmark: _Toc116312605]MCVideo_TransmissionControl_TransmissionRequest
	Reason for change
	New field 'Functional Alias' in 24.581 Table 9.2.4-1 for Rel-17

	Summary of change
	functionalAlias added to MCVideo_TransmissionControl_TransmissionRequest
NOTE: the field is in 36.579-1 Table 5.5.11.1.1-1 already

	TTCN module
	MCVideo_TransmissionControl_TypeDefs, MCVideo_TransmissionControl_Templates



	  type record MCVideo_TransmissionControl_TransmissionRequest {
    TransmissionControl_CommonHdr commonHdr,
    TransmissionControl_TransmissionPriority transmissionPriority optional,
    TransmissionControl_UserID userId optional,
    TransmissionControl_TransmissionIndicator transmissionIndicator optional,
    TransmissionControl_FunctionalAlias functionalAlias optional
  }

	  template (present) MCVideo_TransmissionControlMsg_Type cr_TransmissionControlMsg_TransmissionRequest(template (present) RTP_SSRC p_SSRC := ?,
                                                                                                       template (present) B16_Type p_TransmissionInd) :=
  { // 36.579-1 Table 5.5.11.1.1-1
    transmissionRequest := {
      commonHdr := cr_TransmissionControl_CommonHdr('?0000'B, p_SSRC, tsc_MCVideo_MediaControl_MCV0),
      transmissionPriority := cr_TransmissionControl_TransmissionPriority ifpresent,
      userId := omit,
      transmissionIndicator := cr_TransmissionControl_TransmissionIndicator(p_TransmissionInd),
      functionalAlias := omit
  }



1.2.4.3 [bookmark: _Toc116312606]MCVideo_TransmissionControl_MediaTransmissionNotification
	Reason for change
	New field 'Functional Alias' in 24.581 Table 9.2.13-1 for Rel-17 and new field 'Reception Mode' for Rel-15/16/17

	Summary of change
	functionalAlias and receptionMode added to MCVideo_TransmissionControl_MediaTransmissionNotification 
NOTE 1: the field 'Functional Alias' is in 36.579-1 Table 5.5.11.2.7-1 already
NOTE 2: the field 'Reception Mode' has been added down to Rel-15 of 24.581 and is mentioned even in Rel-14; it seems to be a non-backward compatible change as according to clauses 6.3.6.3.3 and 6.3.6.4.10 the server shall include a Reception Mode field set either to '0' or to '1'  It is assumed that in general the field shall be included in the Media Transmission Notification.

	TTCN module
	MCVideo_TransmissionControl_TypeDefs, MCVideo_TransmissionControl_Templates



	  type record MCVideo_TransmissionControl_MediaTransmissionNotification {
    TransmissionControl_CommonHdr commonHdr,
    TransmissionControl_UserID userId optional,
    TransmissionControl_SSRC ssrcTransmitter optional,
    TransmissionControl_PermissionToRequest permissionToRequest optional,
    TransmissionControl_TransmissionIndicator transmissionIndicator optional,
    TransmissionControl_TrackInfo trackInfo optional,
    TransmissionControl_FunctionalAlias functionalAlias optional,               // Rel-17
    TransmissionControl_ReceptionMode receptionMode optional
  }

	  template (value) TransmissionControl_ReceptionMode cs_TransmissionControl_ReceptionMode(O2_Type p_Val) :=
  {
    id := '16'O,
    lgth := '02'O,
    val := p_Val
  };

	  template (value) MCVideo_TransmissionControlMsg_Type cs_TransmissionControlMsg_MediaTransmissionNotification(B16_Type p_TransmissionInd,
                                                                                                               charstring p_UserID_Transmitter,
                                                                                                               O4_Type p_SSRC_Transmitter,
                                                                                                               O2_Type p_PermissionToRequestTransmissionVal,
                                                                                                               O2_Type p_ReceptionModeVal,
                                                                                                               MCX_MediaControl_AckRequest_Type p_AckRequest := ACK_NOT_REQUIRED) :=
  { // 36.579-1 Table 5.5.11.2.7-1
    mediaTransmissionNotification := {
      commonHdr := cs_TransmissionControl_CommonHdr(f_MCX_MediaControlSubtype_TX('00110'B, p_AckRequest), tsc_MCX_SSRC_NW, tsc_MCVideo_MediaControl_MCV1),
      userId := fl_TransmissionControl_UserID_TX(p_UserID_Transmitter),
      ssrcTransmitter := cs_TransmissionControl_SSRC(p_SSRC_Transmitter),
      permissionToRequest := cs_TransmissionControl_PermissionToRequestTransmission(p_PermissionToRequestTransmissionVal),
      transmissionIndicator := cs_TransmissionControl_TransmissionIndicator(p_TransmissionInd),
      trackInfo := omit,
      functionalAlias := omit,
      receptionMode := cs_TransmissionControl_ReceptionMode(p_ReceptionModeVal)
    }
  };



1.2.4.4 [bookmark: _Toc116312607]MCVideo_TransmissionControl_ReceiveMediaRequest
	Reason for change
	New field 'Functional Alias' in 24.581 Table 9.2.14-1 for Rel-17

	Summary of change
	functionalAlias added to MCVideo_TransmissionControl_ReceiveMediaRequest
NOTE: the field is in 36.579-1 Table 5.5.11.1.4-1 already

	TTCN module
	MCVideo_TransmissionControl_TypeDefs, MCVideo_TransmissionControl_Templates



	  type record MCVideo_TransmissionControl_ReceiveMediaRequest {
    TransmissionControl_CommonHdr commonHdr,
    TransmissionControl_UserID userId optional,
    TransmissionControl_SSRC ssrcTransmitter optional,
    TransmissionControl_TransmissionIndicator transmissionIndicator optional,
    TransmissionControl_ReceptionPriority receptionPriority optional,
    TransmissionControl_TrackInfo trackInfo optional,
    TransmissionControl_FunctionalAlias functionalAlias optional                // Rel-17
  }

	  template (present) MCVideo_TransmissionControlMsg_Type cr_TransmissionControlMsg_ReceiveMediaRequest(template (present) RTP_SSRC p_SSRC,
                                                                                                       charstring p_UserID_Receiver,
                                                                                                       RTP_SSRC p_SSRC_Transmitter,
                                                                                                       template (present) B16_Type p_TransmissionInd) :=
  { // 36.579-1 Table 5.5.11.1.4-1
    receiveMediaRequest := {
      commonHdr := cr_TransmissionControl_CommonHdr('?0100'B, p_SSRC, tsc_MCVideo_MediaControl_MCV0),
      userId := fl_TransmissionControl_UserID_RX(p_UserID_Receiver),
      ssrcTransmitter := cr_TransmissionControl_SSRC(p_SSRC_Transmitter),
      transmissionIndicator := cr_TransmissionControl_TransmissionIndicator(p_TransmissionInd),
      receptionPriority := cr_TransmissionControl_ReceptionPriority ifpresent,
      trackInfo := omit,
      functionalAlias := omit
    }
  };



1.2.4.5 [bookmark: _Toc116312608]MCVideo_TransmissionControl_MediaReceptionNotification
	Reason for change
	New field 'Functional Alias' in 24.581 Table 9.2.16-1 for Rel-17

	Summary of change
	functionalAlias added to MCVideo_TransmissionControl_MediaReceptionNotification
NOTE: the field is in 36.579-1 Table 5.5.11.2.9-1 already

	TTCN module
	MCVideo_TransmissionControl_TypeDefs, MCVideo_TransmissionControl_Templates



	  type record MCVideo_TransmissionControl_MediaReceptionNotification {
    TransmissionControl_CommonHdr commonHdr,
    TransmissionControl_UserID userId optional,
    TransmissionControl_FunctionalAlias functionalAlias optional                // Rel-17
  }

	  template (value) MCVideo_TransmissionControlMsg_Type cs_TransmissionControlMsg_MediaReceptionNotification(MCX_MediaControl_AckRequest_Type p_AckRequest := ACK_NOT_REQUIRED,
                                                                                                            charstring p_UserID_Receiver := px_MCVideo_ID_User_B) :=
  { // 36.579-1 Table 5.5.11.2.9-1
    mediaReceptionNotification := {
      commonHdr := cs_TransmissionControl_CommonHdr(f_MCX_MediaControlSubtype_TX('01000'B, p_AckRequest), tsc_MCX_SSRC_NW, tsc_MCVideo_MediaControl_MCV1),
      userId := fl_TransmissionControl_UserID_TX(p_UserID_Receiver),
      functionalAlias := omit
    }
  }



1.2.5 [bookmark: _Toc116312609]MCData messages (24.282 V17.8.0 clause 15)
1.2.5.1 [bookmark: _Toc116312610]MCData_SDS_SIGNALLING_PAYLOAD, MCData_FD_SIGNALLING_PAYLOAD
	Reason for change
	Change of 24.282 Table 15.1.2.1-1 and Table 15.1.3.1-1

	Summary of change
	Update of TTCN Definitions
NOTE: Change of cs_MCData_SDS_SIGNALLING_PAYLOAD, cr_MCData_SDS_SIGNALLING_PAYLOAD, cs_MCData_FD_SIGNALLING_PAYLOAD and cr_MCData_FD_SIGNALLING_PAYLOAD is subject to RAN5 approval.

	TTCN module
	MCDataMessage_TypeDefs, MCData_MCDataMessages



	  type record MCData_SDS_SIGNALLING_PAYLOAD {                   // Table 15.1.2.1-1
    MCData_MessageType                  messageType,
    MCData_DateAndTime                  dateAndTime,
    MCData_ConversationID               conversationID,
    MCData_MessageID                    messageID,
    MCData_InReplyToMessageID           inReplyTo               optional,
    MCData_ApplicationID                applicationID           optional,
    MCData_SDS_DispositionRequestType   dispositionRequestType  optional,
    MCData_ExtendedApplicationID        extendedApplicationID   optional,
    MCData_UserLocation                 userLocation            optional,
    MCData_UserID                       senderUserID            optional,
    MCData_ApplicationMetadata          applicationMetadata     optional
  };

	  type record MCData_FD_SIGNALLING_PAYLOAD {                    // Table 15.1.3.1-1
    MCData_MessageType                  messageType,
    MCData_DateAndTime                  dateAndTime,
    MCData_ConversationID               conversationID,
    MCData_MessageID                    messageID,
    MCData_InReplyToMessageID           inReplyTo               optional,
    MCData_ApplicationID                applicationID           optional,
    MCData_FD_DispositionRequestType    dispositionRequestType  optional,
    MCData_MandatoryDownload            mandatoryDownload       optional,
    MCData_Payload                      payload                 optional,
    MCData_Metadata                     metadata                optional,
    MCData_ExtendedApplicationID        extendedApplicationID   optional,
    MCData_UserID                       senderUserID            optional,
    MCData_ApplicationMetadata          applicationMetadata     optional
  };

	  type record MCData_ApplicationMetadata  {                     // 24.282 clause 15.2.28
    O1_Type  iei        optional,                               // '53'O
    O2_Type  iel,
    octetstring content
  };

	  template (present) MCData_ApplicationMetadata cr_MCData_ApplicationMetadata(O1_Type p_IEI := '53'O) :=
  {
    iei := p_IEI,
    iel := ?,
    content := ?
  };

	  template (value) MCData_Message cs_MCData_SDS_SIGNALLING_PAYLOAD(integer p_TimestampSeconds,
                                                                   MCData_ConversationID p_ConversationID,
                                                                   MCData_MessageID p_MessageID,
                                                                   H1_Type p_DispositionRequestType) :=
  {
    SDS_SIGNALLING_PAYLOAD := {
      messageType := oct2bit('01'O),
      dateAndTime := int2oct(p_TimestampSeconds, 5),
      conversationID := p_ConversationID,
      messageID := p_MessageID,
      inReplyTo := omit,
      applicationID := omit,
      dispositionRequestType := cs_MCData_SDS_DispositionRequestType(p_DispositionRequestType),
      extendedApplicationID := omit,
      userLocation := omit,
      senderUserID := omit,
      applicationMetadata := omit
    }
  };

	  template (present) MCData_Message cr_MCData_SDS_SIGNALLING_PAYLOAD(template (present) H1_Type p_DispositionRequestType) :=
  {
    SDS_SIGNALLING_PAYLOAD := {
      messageType := oct2bit('01'O),
      dateAndTime := ?,
      conversationID := ?,
      messageID := ?,
      inReplyTo := omit,
      applicationID := omit,
      dispositionRequestType := cr_MCData_SDS_DispositionRequestType(p_DispositionRequestType),
      extendedApplicationID := omit,
      userLocation := *,
      senderUserID := omit,
      applicationMetadata := cr_MCData_ApplicationMetadata ifpresent
    }
  };

	  template (value) MCData_Message cs_MCData_FD_SIGNALLING_PAYLOAD(integer p_TimestampSeconds,
                                                                  MCData_ConversationID p_ConversationID,
                                                                  MCData_MessageID p_MessageID,
                                                                  template (omit) MCData_FD_DispositionRequestType p_DispositionRequestType,
                                                                  template (omit) MCData_MandatoryDownload p_MandatoryDownload := omit,
                                                                  template (omit) MCData_Payload p_Payload := omit,
                                                                  template (omit) MCData_Metadata p_Metadata := omit) :=
  {
    FD_SIGNALLING_PAYLOAD := {
      messageType := oct2bit('02'O),
      dateAndTime := int2oct(p_TimestampSeconds, 5),
      conversationID := p_ConversationID,
      messageID := p_MessageID,
      inReplyTo := omit,
      applicationID := omit,
      dispositionRequestType := p_DispositionRequestType,
      mandatoryDownload := p_MandatoryDownload,
      payload := p_Payload,
      metadata := p_Metadata,
      extendedApplicationID := omit,
      senderUserID := omit,
      applicationMetadata := omit
    }
  };

	  template (present) MCData_Message cr_MCData_FD_SIGNALLING_PAYLOAD(template MCData_FD_DispositionRequestType p_DispositionRequestType := cr_MCData_FD_DispositionRequestType,
                                                                    template MCData_MandatoryDownload p_MandatoryDownload := omit,
                                                                    template MCData_Payload p_Payload := omit,
                                                                    template MCData_Metadata p_Metadata := omit) :=
  {
    FD_SIGNALLING_PAYLOAD := {
      messageType := oct2bit('02'O),
      dateAndTime := ?,
      conversationID := ?,
      messageID := ?,
      inReplyTo := omit,
      applicationID := omit,
      dispositionRequestType := p_DispositionRequestType,
      mandatoryDownload := p_MandatoryDownload,
      payload := p_Payload,
      metadata := p_Metadata,
      extendedApplicationID := omit,
      senderUserID := omit,
      applicationMetadata := cr_MCData_ApplicationMetadata ifpresent
    }



1.2.5.2 [bookmark: _Toc116312611]MCData_DEFERRED_DATA_RESPONSE
	Reason for change
	Change of 24.282 Table 15.1.12.1-1

	Summary of change
	Update of TTCN Definitions
NOTE: Change of cs_MCData_DEFERRED_DATA_RESPONSE is subject to RAN5 approval.

	TTCN module
	MCDataMessage_TypeDefs, MCData_MCDataMessages

	MCC160 Comment
	



	  type record MCData_DEFERRED_DATA_RESPONSE {                   // Table 15.1.12.1-1
    MCData_MessageType                  messageType,
    MCData_NumberOfPayloads             numberOfPayloads,
    MCData_ProtectedPayloadMessageList  protectedPayloads,
    MCData_PayloadList                  payloads,
    MCData_GroupID                      groupID                 optional,
    MCData_DeferredPayloadList          deferredPayloads
  };

	  type record MCData_DeferredPayload {                          // 24.282 clause 15.2.27
    O1_Type  iei        optional,                               // '52'O
    O2_Type  iel,
    octetstring content
  };

	  type record of MCData_DeferredPayload MCData_DeferredPayloadList;

	  template (value) MCData_Message cs_MCData_DEFERRED_DATA_RESPONSE(template (value) MCData_Payload p_Payload) :=
  {
    DEFERRED_DATA_RESPONSE := {
      messageType := oct2bit('0C'O),
      numberOfPayloads := '01'O,
      protectedPayloads := {},
      payloads := { p_Payload },
      groupID := omit
      deferredPayloads := {}
    }
  };



