
	Reason for change:
	1. At step 5-6 when Scell 3 is removed and replaced with Scell 12, sCellToReleaseList  with ScellIndex 2 should be included in RRCReconfiguration.

2. At step 7-8, when Scell 12 is made PCell with no SCell, sCellToReleaseList  with ScellIndex 2 should be included in RRCReconfiguration.

3. The sCellIndex of NR cell 12 is ‘2 ‘ in RRCReconfiguration at step 5-6 but this is not configured to SS.


4. During inter-cell handover security key is changed due to reconfiguration with Sync, re-establishPDCP set to true and reestablishRLC has to be set to true

5. If “p_SCGConfig.sCellToAddModList” is not present, the function _NR_ConfigSCell() will give a run-time error when attempting to access “SCGConfig.sCellToAddModList[0].sCellConfigDedicated.tag_Id”. This happens at step 7 where this function is called through f_NR_RRC_PSCellChangeCA_InterCell() but sCellToAddModList is not present.


6. If IP ping is not supported, UE should have test loop mode B activated in the preamble and open for the data path check at step 3A and 6A, and closed again before postamble sequence. 

7. Missing coordaintion message coordination message to trigger change power in NR cell at step 1

8. The function f_NR_SS_SRBs_DRBs_Release() is called through f_NR_ReconfigurePCellToNormalCell() and is not previously used for EN-DC test cases, and now needs to take into account that SRBs will not have been configured on the SS NR cell in these scenarios.



	
	

	Summary of change:
	1. sCellToReleaseList set with ScellIndex 2 in secondary cell group config.

2. sCellToReleaseList set with ScellIndex 2 in secondary cell group config.

3. Update SS with new SCell configuration

4. Set reestablishRLC and call  f_NR_RetrieveRadioBearerConfig_PSCellChange invoked with key change paramter set true. 

5. Added a check for SCGConfig.sCellToAddModList being present before accessing

6. Update preamble condition and test loop mode B is opened before step 3A and closed after step 8

7. Added the coordination message for ‘change Power’

8. SRB1 and SRB2 not added to the release list in the case of EN-DC.

Note : An associated prose CR  on38.523-1 will be submitted at RAN5#96

1. [bookmark: _Toc122434485][bookmark: _Toc109936065]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D22wk23 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc109936066]Corrections required

1. [bookmark: _Toc30685521][bookmark: _Toc109936067]Change 1
	Function name
	fl_TC_8_2_4_2_1_x_ENDC_NR_TestBody()

	Reason for change
	1. At step 5-6 when Scell 3 is removed and replaced with Scell 12, sCellToReleaseList  with ScellIndex 2 should be included in RRCReconfiguration.
2. At step 7-8, when Scell 12 is made PCell with no SCell, sCellToReleaseList  with ScellIndex 2 should be included in RRCReconfiguration.
3. The sCellIndex of NR cell 12 is ‘2 ‘ in RRCReconfiguration at step 5-6 but this is not configured to SS.
4. During inter-cell handover security key is changed due to reconfiguration with Sync, re-establishPDCP set to true and reestablishRLC has to be set to true

	Summary of change
	1. sCellToReleaseList set with ScellIndex 2 in secondary cell group config.
2. sCellToReleaseList set with ScellIndex 2 in secondary cell group config.
3. Update SS with new SCell configuration
4. f_NR_RetrieveRadioBearerConfig_PSCellChange invoked with first argument as true
Note : An associated prose CR  on38.523-1 will be submitted at RAN5#96

	TTCN module
	RRC_CarrierAggregation_ENDC_NR

	MCC160 Comment
	



Before Change
	function fl_TC_8_2_4_2_1_x_ENDC_NR_TestBody(NR_CellId_Type  p_PSCell1,
                                              NR_CellId_Type  p_PSCell2,
                                              NR_CellId_Type  p_SCell1,
                                              NR_CellId_Type  p_SCell2) runs on ENDC_NR_PTC
  {
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) CellGroupConfig v_SCGConfig := f_NR_GetSCGConfigDef(p_PSCell2);
    var template(value) CellGroupConfig v_SCGConfig_SCell := f_NR_GetSCGConfig_SCellAdd(p_SCell2, 2);
    var template (value) RadioBearerConfig v_RadioBearerConfig := cs_NR_RadioBearer(tsc_NR_DRB2, secondary, f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get());
    var template (value) NR_RadioBearerList_Type v_RadioBearerToReleaseList := {cs_NR_SS_DRB_Release(tsc_NR_DRB2)};
    
    v_CellPowerList_AtT0 := {cs_NR_CellPower(p_PSCell1, -88, -91),
                             cs_NR_CellPower(p_SCell1, -88, -91)};
    v_CellPowerList_AtT1 := {cs_NR_CellPower(p_PSCell2, -88, -91),
                             cs_NR_CellPower(p_SCell2, -88, -91)};

    // @siclog "Step 1" siclog@
    //The SS changes Cell parameters according to the row "T0" in table 8.2.4.2.1.1.3.2-1.
    // Send trigger to NR to change power level of NR Cell1
    f_NR_WaitForEUTRATrigger_And_ChangePower(v_CellPowerList_AtT0);

    //@siclog "Steps 2 - 3" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to configure NR Cell 1 as the NR PSCell and sCellToAddModList with an sCellIndex set to NR Cell 3 as the configured SCell.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?
    f_NR_AddMod_PSCell_SCell_ENDC( p_PSCell1, p_SCell1 );

    //@siclog "Step 3A" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1A indicate that the UE is capable of exchanging IP data on DRB #n associated with the first PDU session on NR Cell 3?
    //@sic R5-203666 sic@
    f_NR_CheckDataPath_CA(p_PSCell1, p_SCell1, "3A", true,tsc_SCellIndex_2);
    
    //@siclog "Step 4" siclog@
    //The SS changes NR Cell 2 and NR Cell 12 parameters according to the row "T1" in table 8.2.4.2.1.1.3.2-1.
    f_NR_WaitForEUTRATrigger_And_ChangePower(v_CellPowerList_AtT1);
        
    //@siclog "Steps 5 - 6" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to perform both NR Cell 1 and NR Cell 3 changed to the configured target NR Cell 2 and NR Cell 12.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message configuring the new PSCell and SCell and configure lower layers to consider the SCell to be in deactivated state?
    v_SCGConfig.sCellToAddModList := v_SCGConfig_SCell.sCellToAddModList;
    v_RadioBearerConfig := f_NR_RetrieveRadioBearerConfig_PSCellChange( false, -, secondary);
    f_NR_RRC_PSCellChangeCA_InterCell(p_PSCell1, p_SCell1, p_PSCell2, p_SCell2, false, v_RadioBearerConfig, v_SCGConfig, v_RadioBearerToReleaseList);
    
    //@siclog "Step 6A" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1A indicate that the UE is capable of exchanging IP data on DRB #n associated with the first PDU session on NR Cell 12?
    //@sic R5-203666 sic@
    f_NR_CheckDataPath_CA(p_PSCell2, p_SCell2, "6A", true, tsc_SCellIndex_2);

    //@siclog "Steps 7 - 8" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to change the current PSCell (NR Cell 2) to NR Cell 12 equaling to one of the current UE SCell configuration.
    //Check: Does the UE send an RRCConnectionReconfigurationComplete message and reconfigure the current NR SCell to PSCell?
    v_SCGConfig := f_NR_GetSCGConfigDef(p_SCell2);
    v_SCGConfig_SCell := f_NR_GetSCGConfigDef(p_PSCell2);
    v_SCGConfig.sCellToAddModList := v_SCGConfig_SCell.sCellToAddModList;
    f_NR_ReconfigSCellToNormalCell(p_SCell2,  {f_NR_GetRadioBearerConfig_FullAM_DRB2(p_PSCell2)});
    f_NR_RRC_PSCellChangeCA_InterCell(p_PSCell2, p_SCell2, p_SCell2, p_PSCell2, false, v_RadioBearerConfig, v_SCGConfig, v_RadioBearerToReleaseList);
    f_NR_ReconfigurePCellToNormalCell( p_PSCell2, tsc_PCellIndex_1, -, {tsc_NR_DRB2} );
  }



After Change
	function fl_TC_8_2_4_2_1_x_ENDC_NR_TestBody(NR_CellId_Type  p_PSCell1,
                                              NR_CellId_Type  p_PSCell2,
                                              NR_CellId_Type  p_SCell1,
                                              NR_CellId_Type  p_SCell2) runs on ENDC_NR_PTC
  {
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) CellGroupConfig v_SCGConfig := f_NR_GetSCGConfigDef(p_PSCell2);
    var template(value) CellGroupConfig v_SCGConfig_SCell := f_NR_GetSCGConfig_SCellAdd(p_SCell2, 2);
    var template (value) RadioBearerConfig v_RadioBearerConfig := cs_NR_RadioBearer(tsc_NR_DRB2, secondary, f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get());
    var template (value) NR_RadioBearerList_Type v_RadioBearerToReleaseList := {cs_NR_SS_DRB_Release(tsc_NR_DRB2)};
    var template (omit) TAG_Id v_TAG_Id ; 
    v_CellPowerList_AtT0 := {cs_NR_CellPower(p_PSCell1, -88, -91),
                             cs_NR_CellPower(p_SCell1, -88, -91)};
    v_CellPowerList_AtT1 := {cs_NR_CellPower(p_PSCell2, -88, -91),
                             cs_NR_CellPower(p_SCell2, -88, -91)};

    // @siclog "Step 1" siclog@
    //The SS changes Cell parameters according to the row "T0" in table 8.2.4.2.1.1.3.2-1.
    // Send trigger to NR to change power level of NR Cell1
    f_NR_WaitForEUTRATrigger_And_ChangePower(v_CellPowerList_AtT0);

    //@siclog "Steps 2 - 3" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to configure NR Cell 1 as the NR PSCell and sCellToAddModList with an sCellIndex set to NR Cell 3 as the configured SCell.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?
    f_NR_AddMod_PSCell_SCell_ENDC( p_PSCell1, p_SCell1 );
    //Get security info from EUTRA,calculate latest keys and configure SS
    f_ENDC_GetKeysActivateSecurity(p_SCell1,EUTRA);
    //@siclog "Step 3A" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1A indicate that the UE is capable of exchanging IP data on DRB #n associated with the first PDU session on NR Cell 3?
    //@sic R5-203666 sic@
    f_NR_CheckDataPath_CA(p_PSCell1, p_SCell1, "3A", true,tsc_SCellIndex_2);
    
    //@siclog "Step 4" siclog@
    //The SS changes NR Cell 2 and NR Cell 12 parameters according to the row "T1" in table 8.2.4.2.1.1.3.2-1.
    f_NR_WaitForEUTRATrigger_And_ChangePower(v_CellPowerList_AtT1);
        
    //@siclog "Steps 5 - 6" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to perform both NR Cell 1 and NR Cell 3 changed to the configured target NR Cell 2 and NR Cell 12.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message configuring the new PSCell and SCell and configure lower layers to consider the SCell to be in deactivated state?
    
    v_SCGConfig.sCellToAddModList := v_SCGConfig_SCell.sCellToAddModList;
     f_NR_ConfigSCell(p_SCell2,v_SCGConfig,cs_TimingInfo_Now);

   v_SCGConfig.rlc_BearerToAddModList[0].reestablishRLC := true_;
    
    v_SCGConfig.sCellToReleaseList[0] := tsc_SCellIndex_2;
    v_RadioBearerConfig := f_NR_RetrieveRadioBearerConfig_PSCellChange( true, -, secondary);

    f_NR_RRC_PSCellChangeCA_InterCell(p_PSCell1, p_SCell1, p_PSCell2, p_SCell2, false, v_RadioBearerConfig, v_SCGConfig, v_RadioBearerToReleaseList);
    
    //@siclog "Step 6A" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1A indicate that the UE is capable of exchanging IP data on DRB #n associated with the first PDU session on NR Cell 12?
    //@sic R5-203666 sic@
    f_NR_CheckDataPath_CA(p_PSCell2, p_SCell2, "6A", true, tsc_SCellIndex_2);

    //@siclog "Steps 7 - 8" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to change the current PSCell (NR Cell 2) to NR Cell 12 equaling to one of the current UE SCell configuration.
    //Check: Does the UE send an RRCConnectionReconfigurationComplete message and reconfigure the current NR SCell to PSCell?
    v_SCGConfig := f_NR_GetSCGConfigDef(p_SCell2);
    v_SCGConfig_SCell := f_NR_GetSCGConfigDef(p_PSCell2);

     v_SCGConfig.rlc_BearerToAddModList[0].reestablishRLC := true_;
    v_SCGConfig.sCellToAddModList := v_SCGConfig_SCell.sCellToAddModList;
    f_NR_ReconfigSCellToNormalCell(p_SCell2,  {f_NR_GetRadioBearerConfig_FullAM_DRB2(p_PSCell2)});
    
    v_SCGConfig.sCellToReleaseList[0] := tsc_SCellIndex_2;
    f_NR_RRC_PSCellChangeCA_InterCell(p_PSCell2, p_SCell2, p_SCell2, p_PSCell2, false, v_RadioBearerConfig, v_SCGConfig, v_RadioBearerToReleaseList);
    f_NR_ReconfigurePCellToNormalCell( p_PSCell2, tsc_PCellIndex_1, -, {tsc_NR_DRB2} );
  }
  


1. [bookmark: _Toc109936068][bookmark: _Toc30685522]Change 2
	Function name
	f_NR_ConfigSCell()

	Reason for change
	If “p_SCGConfig.sCellToAddModList” is not present, this function will give a run-time error when attempting to access “SCGConfig.sCellToAddModList[0].sCellConfigDedicated.tag_Id”. This happens at step 7 where this function is called through f_NR_RRC_PSCellChangeCA_InterCell() but sCellToAddModList is not present.


	Summary of change
	Added a check for SCGConfig.sCellToAddModList being present or not before accessing 

	TTCN module
	NR_CommonFunctions_CA

	MCC160 Comment
	



Before Change
	function f_NR_ConfigSCell(NR_CellId_Type p_SCellId,
                                   template(value) CellGroupConfig p_SCGConfig,
                                   template(value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now
                                   ) runs on NR_BASE_PTC
  {
    var template (omit) TAG_Id v_TAG_Id :=  p_SCGConfig.sCellToAddModList[0].sCellConfigDedicated.tag_Id;
    var template (value) NR_ServingCellIndex_Type v_SCellIndex :=  p_SCGConfig.sCellToAddModList[0].sCellIndex;

    //other PCell reconfiguration depending on the TC
    // CQI report
    //@sic R5s200462 change 2 sic@
    
    // Configure SCell as Scell
    f_NR_SS_CommonCellConfig(p_SCellId, cads_NR_ServingCellConfig_REQ(p_SCellId, p_TimingInfo,  //@sic R5s200462 change 2 sic@
                                                                      cs_NR_ServingCellConfig_SCell(v_SCellIndex, v_TAG_Id)));
                                                                      
   
  }



After Change
	function f_NR_ConfigSCell(NR_CellId_Type p_SCellId,
                                   template(value) CellGroupConfig p_SCGConfig,
                                   template(value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now
                                   ) runs on NR_BASE_PTC
  {
  if(ispresent(p_SCGConfig.sCellToAddModList))
  {
    var template (omit) TAG_Id v_TAG_Id :=  p_SCGConfig.sCellToAddModList[0].sCellConfigDedicated.tag_Id;
    var template (value) NR_ServingCellIndex_Type v_SCellIndex :=  p_SCGConfig.sCellToAddModList[0].sCellIndex;

    //other PCell reconfiguration depending on the TC
    // CQI report
    //@sic R5s200462 change 2 sic@
    
    // Configure SCell as Scell
    f_NR_SS_CommonCellConfig(p_SCellId, cads_NR_ServingCellConfig_REQ(p_SCellId, p_TimingInfo,  //@sic R5s200462 change 2 sic@
                                                                      cs_NR_ServingCellConfig_SCell(v_SCellIndex, v_TAG_Id)));
                                                                      
   }
   
  }




1. [bookmark: _Toc109936069]Change 3
	Function name
	f_TC_8_2_4_2_1_x_EUTRA_Common()

	Reason for change
	If IP ping is not supported, UE should have test loop mode B activated for the data path check at step 3A and 6A

	Summary of change
	p_TestModeState  is set to PING_Or_TESTModeB_ON in EUTRA preamble

	TTCN module
	RRC_CarrierAggregation_ENDC_EUTRA

	MCC160 Comment
	



Before Change
	  function f_TC_8_2_4_2_1_x_EUTRA_Common() runs on EUTRA_5GS_PTC {
    // SCG Change / Reconfiguration with sync / Split DRB / EN-DC
    f_EUTRA38_Init_CA(EN_DC, c1);

    //Create and configure cell
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    
    //Preamble to enter UE into RRC Connected
    //The UE is in state RRC_CONNECTED using generic procedure parameter Connectivity (EN-DC) according to TS 38.508-1 [4].
   f_EUTRA38_ENDC_Preamble (eutra_Cell1, TESTMode_OFF, MCG_Only, RRC_CONNECTED);
    f_EUTRA_TestBody_Set(true);
    fl_TC_8_2_4_2_1_x_ENDC_EUTRA_TestBody ();
    f_EUTRA_TestBody_Set(false);

    f_EUTRA38_ENDC_Postamble(eutra_Cell1, E2_CONNECTED); //@sic R5-204475 sic@
    // If finishing in E2_CONNECTED, need to wait until connection released before taking down the NR cell
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));
  }



After Change
	    function f_TC_8_2_4_2_1_x_EUTRA_Common() runs on EUTRA_5GS_PTC {
    // SCG Change / Reconfiguration with sync / Split DRB / EN-DC
    f_EUTRA38_Init_CA(EN_DC, c1);

    //Create and configure cell
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    
    //Preamble to enter UE into RRC Connected
    //The UE is in state RRC_CONNECTED using generic procedure parameter Connectivity (EN-DC) according to TS 38.508-1 [4].
    f_EUTRA38_ENDC_Preamble (eutra_Cell1, PING_Or_TESTModeB_ON, MCG_Only, RRC_CONNECTED);  

    f_EUTRA_TestBody_Set(true);
    fl_TC_8_2_4_2_1_x_ENDC_EUTRA_TestBody ();
    f_EUTRA_TestBody_Set(false);

    f_EUTRA38_ENDC_Postamble(eutra_Cell1, E2_CONNECTED); //@sic R5-204475 sic@
    // If finishing in E2_CONNECTED, need to wait until connection released before taking down the NR cell
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));
  } 



1. [bookmark: _Toc109936070]Change 4
	Function name
	fl_TC_8_2_4_2_1_x_ENDC_EUTRA_TestBody()

	Reason for change
	1. If IP ping is not supported, UE should have test loop mode B open for the data path check at step 3A and 6A, and vlosed again before postamble sequence. 
2. Missing coordaintion message coordination message to trigger change power in NR cell at step 1


	Summary of change
	1. Test loop mode B is open before step 3A and closed after step 8
2. Added the coordination message for ‘change Power’

	TTCN module
	RRC_CarrierAggregation_ENDC_EUTRA

	MCC160 Comment
	



	  function fl_TC_8_2_4_2_1_x_ENDC_EUTRA_TestBody() runs on EUTRA_5GS_PTC
  {
    var EUTRA38_SecurityParamsExtd_Type v_Security_Params;
    
    // @siclog "Step 1" siclog@
    //The SS changes Cell parameters according to the row "T0" in table 8.2.4.2.1.1.3.2-1.
    // Send trigger to NR to change power level of NR Cell1
    f_EUTRA38_SetCellPower(eutra_Cell1, -85, -96);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);
    
    //@siclog "Steps 2 - 3" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to configure NR Cell 1 as the NR PSCell and sCellToAddModList with an sCellIndex set to NR Cell 3 as the configured SCell.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?
    // Initiate SgNB Keys @sic R5s200647 sic@
    v_Security_Params := f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR);
    f_EUTRA38_Security_Set(v_Security_Params);
    f_EUTRA38_508_ENDC_Reconfiguration (eutra_Cell1, MCG_SCG, v_Security_Params.SK_Counter);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");
    
    //@siclog "Step 4" siclog@
    //The SS changes NR Cell 2 and NR Cell 12 parameters according to the row "T1" in table 8.2.4.2.1.1.3.2-1.
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);
        
    //@siclog "Steps 5 - 6" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to perform both NR Cell 1 and NR Cell 3 changed to the configured target NR Cell 2 and NR Cell 12.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message configuring the new PSCell and SCell and configure lower layers to consider the SCell to be in deactivated state?
    v_Security_Params := f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR, 1);
    f_EUTRA38_RRC_PSCellChange_InterCell (eutra_Cell1, v_Security_Params.SK_Counter);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    
    //@siclog "Steps 7 - 8" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to change the current PSCell (NR Cell 2) to NR Cell 12 equaling to one of the current UE SCell configuration.
    v_Security_Params := f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR, 1);
    f_EUTRA38_RRC_PSCellChange_InterCell (eutra_Cell1, v_Security_Params.SK_Counter);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");
  }



After Change

	  function fl_TC_8_2_4_2_1_x_ENDC_EUTRA_TestBody() runs on EUTRA_5GS_PTC
  {
    var EUTRA38_SecurityParamsExtd_Type v_Security_Params;
    
    // @siclog "Step 1" siclog@
    //The SS changes Cell parameters according to the row "T0" in table 8.2.4.2.1.1.3.2-1.
    // Send trigger to NR to change power level of NR Cell1
    f_EUTRA38_SetCellPower(eutra_Cell1, -85, -96);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("ChangePower"));
    
    //@siclog "Steps 2 - 3" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to configure NR Cell 1 as the NR PSCell and sCellToAddModList with an sCellIndex set to NR Cell 3 as the configured SCell.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?
    // Initiate SgNB Keys @sic R5s200647 sic@
    v_Security_Params := f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR);
    f_EUTRA38_Security_Set(v_Security_Params);
    f_EUTRA38_508_ENDC_Reconfiguration (eutra_Cell1, MCG_SCG, v_Security_Params.SK_Counter);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");
    if(not pc_IP_Ping)
    {
      f_EUTRA_CloseUE_TestLoopModeB(eutra_Cell1,'00'O);
    }
    //Initiate SgNB Keys @sic R5s200647 sic@
    
    v_Security_Params := f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR);
    
    f_EUTRA38_Security_Set(v_Security_Params);
    
    //@siclog "Step 4" siclog@
    //The SS changes NR Cell 2 and NR Cell 12 parameters according to the row "T1" in table 8.2.4.2.1.1.3.2-1.
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("ChangePower"));
     
    //@siclog "Steps 5 - 6" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to perform both NR Cell 1 and NR Cell 3 changed to the configured target NR Cell 2 and NR Cell 12.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message configuring the new PSCell and SCell and configure lower layers to consider the SCell to be in deactivated state?
    v_Security_Params := f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR, 1);
    f_EUTRA38_RRC_PSCellChange_InterCell (eutra_Cell1, v_Security_Params.SK_Counter);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    
    //@siclog "Steps 7 - 8" siclog@
    //The SS transmits an RRCConnectionReconfiguration message with nr-Config IE including NR RRCReconfiguration message to change the current PSCell (NR Cell 2) to NR Cell 12 equaling to one of the current UE SCell configuration.
    v_Security_Params := f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR, 1);
    f_EUTRA38_RRC_PSCellChange_InterCell (eutra_Cell1, v_Security_Params.SK_Counter);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");
      if (not pc_IP_Ping) {       //@sic R5s201471 R5-205577 sic@
      f_EUTRA_OpenUE_TestLoopMode_Deactivate_TestMode (eutra_Cell1);
    }
  }



1. [bookmark: _Toc109936071]Change 5
	Function name
	f_NR_SS_SRBs_DRBs_Release()

	Reason for change
	This function is called through f_NR_ReconfigurePCellToNormalCell() and is not previously used for EN-DC test cases, and now needs to take into account that SRBs will not have been configured on the SS NR cell in these scenarios.

	Summary of change
	SRB1 and SRB2 not added to the release list in the case of EN-DC.

	TTCN module
	NR_ConfigurationSteps

	MCC160 Comment
	



	function f_NR_SS_SRBs_DRBs_Release(NR_CellId_Type  p_NR_CellId,
                                     template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                     IntegerList_Type p_DrbList,
                                     template (omit) boolean p_CnfFlag := omit,
                                     boolean p_FollowOnFlag := tsc_NoFollowOnFlag
                                     ) runs on NR_BASE_PTC
  {
    var template (value) NR_RadioBearerList_Type v_SrbDrbList := {
      cs_NR_SS_RadioBearer_Release(tsc_NR_RbId_SRB1),
      cs_NR_SS_RadioBearer_Release(tsc_NR_RbId_SRB2)
    };
    var integer i := lengthof(v_SrbDrbList);
    var integer k;
    
    for (k:=0; k < lengthof(p_DrbList); k:= k+1) {
      v_SrbDrbList[i+k] := cs_NR_SS_DRB_Release(p_DrbList[k]);
    }
    f_NR_SS_CommonRadioBearerConfig(p_NR_CellId, v_SrbDrbList, p_TimingInfo, p_CnfFlag, p_FollowOnFlag);
  }



After Change

	  function f_NR_SS_SRBs_DRBs_Release(NR_CellId_Type  p_NR_CellId,
                                     template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                     IntegerList_Type p_DrbList,
                                     template (omit) boolean p_CnfFlag := omit,
                                     boolean p_FollowOnFlag := tsc_NoFollowOnFlag
                                     ) runs on NR_BASE_PTC
  {
    var template (value) NR_RadioBearerList_Type v_SrbDrbList := {};
    var integer i;
    var integer k;
    
    if(not f_GetTestcaseAttrib_ENDC(testcasename()))
    {
    v_SrbDrbList := {

      cs_NR_SS_RadioBearer_Release(tsc_NR_RbId_SRB1),

      cs_NR_SS_RadioBearer_Release(tsc_NR_RbId_SRB2)
    };
      i := lengthof(v_SrbDrbList);
       for (k:=0; k < lengthof(p_DrbList); k:= k+1) {
      v_SrbDrbList[i+k] := cs_NR_SS_DRB_Release(p_DrbList[k]);
    }
      
     f_NR_SS_CommonRadioBearerConfig(p_NR_CellId, v_SrbDrbList, p_TimingInfo, p_CnfFlag, p_FollowOnFlag);
    }
  }








