
	Reason for change:
	Some corrections required following the implementation of R5-223463 and R5s220575.

1. Following implementation of R5-223463, in the 2nd call setup sequence of 8.34/8.35, the function f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7() will call f_IMS_MOCallSetup_ReceivePRACK_Send200OK() with an expected updated SDP offer template in the parameter p_SdpTemplatePRACK (because p_RadioBearersEstablished will be true in f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7()). 

It is still possible that the PRACK from the UE will not include an SDP body as the UE may provide it in UPDATE later. So in this case the function f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP_Ifpresent() will set a failure verdict due to non-matching SDP content.


2. Following implementation of R5-223463, UE may not include SDP content in the PRACK message during the 2nd and 3rd MO call setup sequences, and so content type header may not be included.

3. When two calls are merged and the 3rd dialog to the conference factory is created, the function f_IMS_ConferenceCreation_NR5GC_A_19_Steps1_9() sets the parameter p_EarlyQosConfirmation to false when calling f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7(). With this parameter set false the content type header is not allowed in PRACK message, but the UE may still include SDP content in this message.

4. Following implementation of MCC comments for R5s220575, the checking of url parameters in the refer-to header is now too strict - the checking confirms that in the case of 8.36 the “replaces” and “require” headers are present (and no other header) but it now does not allow the header to be present in other cases. In other cases as e.g. 8.34/8.35, the UE may still include the “require” header along with the “replaces” header in the refer-to URI.






	
	

	Summary of change:
	1. If no SDP body is present in the PRACK message, the function f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP_Ifpresent() is called with omit instead of the SDP content which was passed from f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7()

2. Add ifpresent to the expected content type header 

3. Changed the parameter v_EarlyQosConfirmation value to false

4. After adding the mandatory headers as per MCC implementation of R5s220575, add a “superset” operation to allow inclusion of other parameters (similar to the original proposal of R5s220575).


	
	

	Consequences if not approved:
	A conformant UE can fail some test cases

1. Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D22wk23 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
2. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc112156484]Corrections required
2.1 [bookmark: _Toc107254009][bookmark: _Toc112156485]Change 1
	Function name
	f_IMS_MOCallSetup_ReceivePRACK_Send200OK

	Reason for change
	Following implementation of R5-223463, in the 2nd call setup sequence of 8.34/8.35, the function f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7() will call f_IMS_MOCallSetup_ReceivePRACK_Send200OK() with an expected updated SDP offer template in the parameter p_SdpTemplatePRACK (because p_RadioBearersEstablished will be true in f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7()). 
It is still possible that the PRACK from the UE will not include an SDP body as the UE may provide it in UPDATE later. So in this case the function f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP_Ifpresent() will set a failure verdict due to non-matching SDP content.

	Summary of change
	If no SDP body is present in the PRACK message, the function f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP_Ifpresent() is called with omit instead of the SDP content which was passed from f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7()

	TTCN module
	IMS_Procedures_CallControl4G5G.ttcn

	MCC160 Comment
	



Before Change:
	    function f_IMS_MOCallSetup_ReceivePRACK_Send200OK(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,
                                                    template (value) MessageHeader p_MessageHeader_PrevReliableResponse,
                                                    template ContentType p_ContentTypePRACK := omit,
                                                    template SDP_Message p_SdpTemplatePRACK := omit,
                                                    IMS_CallType_Type p_CallType := NORMAL_CALL) runs on IMS_PTC return template (omit) SDP_Message
  { /* @sic R5s201620: p_ContactUrl removed sic@ */
    /* @sic R5-211439: Handling of SDP sic@ */
    /* @sic R5-213531, R5-214884: p_CallType sic@ */
    /* @sic R5-223463: return SDP_Message - SDP offer in PRACK sic@ */
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var template (present) MessageHeader v_MessageHeaderPRACK;
    var template (present) SipUrl v_ContactUrl;
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var PRACK_Request v_PrackRequest;
    var template (value) MessageHeader v_MessageHeader200OK;
    var template (omit) MessageBody v_MessageBody200OK := omit;
    var template (omit) ContentType v_ContentType200OK := omit;
    var boolean v_PreconditionsRequired200OK := false;
    var template (omit) SDP_Message v_SdpOffer;
    var template (value) SDP_Message v_SdpAnswer;

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_MessageHeader_PrevReliableResponse);     /* @sic R5s201620: ContactUrl retrieved from response being sent reliable sic@ */
    v_MessageHeaderPRACK := f_IMS_PrackRequest_MessageHeaderRX(p_InviteRequestWithSdp.Invite, p_MessageHeader_PrevReliableResponse, p_ContentTypePRACK);

    v_IMS_DATA_REQ := f_IMS_ReceiveRequest(car_IMS_Prack_Request(cr_PRACK_Request(v_ContactUrl, v_MessageHeaderPRACK)));
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);                   /* @sic R5s201620: RoutingInfo derived from PRACK but not from INVITE sic@ */
    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;
    
    if (isvalue(v_PrackRequest.messageBody) xor isvalue(v_PrackRequest.msgHeader.contentType)) {
      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Either the ContentType or the MessageBody is missing");
    }
    v_SdpOffer := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP_Ifpresent(v_PrackRequest.messageBody, p_SdpTemplatePRACK);
    if (isvalue(v_SdpOffer)) {
      f_IMS_MessageHeader_CheckPreconditionIndicationForOffer(v_PrackRequest.msgHeader, PreconditionRequired);     /* check that PRACK contains Require with "preconditions" */
      v_SdpAnswer := f_IMS_BuildSDP_UpdatedAnswer_TX(valueof(v_SdpOffer));
      v_MessageBody200OK := cs_MessageBody_SDP(v_SdpAnswer);
      v_ContentType200OK  := cs_ContentTypeSDP;
      v_PreconditionsRequired200OK := true;
    }

………………………….
  }




After Change:
	    function f_IMS_MOCallSetup_ReceivePRACK_Send200OK(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,
                                                    template (value) MessageHeader p_MessageHeader_PrevReliableResponse,
                                                    template ContentType p_ContentTypePRACK := omit,
                                                    template SDP_Message p_SdpTemplatePRACK := omit,
                                                    IMS_CallType_Type p_CallType := NORMAL_CALL) runs on IMS_PTC return template (omit) SDP_Message
  { /* @sic R5s201620: p_ContactUrl removed sic@ */
    /* @sic R5-211439: Handling of SDP sic@ */
    /* @sic R5-213531, R5-214884: p_CallType sic@ */
    /* @sic R5-223463: return SDP_Message - SDP offer in PRACK sic@ */
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var template (present) MessageHeader v_MessageHeaderPRACK;
    var template (present) SipUrl v_ContactUrl;
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var PRACK_Request v_PrackRequest;
    var template (value) MessageHeader v_MessageHeader200OK;
    var template (omit) MessageBody v_MessageBody200OK := omit;
    var template (omit) ContentType v_ContentType200OK := omit;
    var boolean v_PreconditionsRequired200OK := false;
    var template (omit) SDP_Message v_SdpOffer;
    var template (value) SDP_Message v_SdpAnswer;
    var template (omit) SDP_Message v_SdpTemplatePRACK := omit;

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_MessageHeader_PrevReliableResponse);     /* @sic R5s201620: ContactUrl retrieved from response being sent reliable sic@ */
    v_MessageHeaderPRACK := f_IMS_PrackRequest_MessageHeaderRX(p_InviteRequestWithSdp.Invite, p_MessageHeader_PrevReliableResponse, p_ContentTypePRACK);

    v_IMS_DATA_REQ := f_IMS_ReceiveRequest(car_IMS_Prack_Request(cr_PRACK_Request(v_ContactUrl, v_MessageHeaderPRACK)));
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);                   /* @sic R5s201620: RoutingInfo derived from PRACK but not from INVITE sic@ */
    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;
    
    if (isvalue(v_PrackRequest.messageBody) xor isvalue(v_PrackRequest.msgHeader.contentType)) {
      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Either the ContentType or the MessageBody is missing");
    }
    if (isvalue(v_PrackRequest.messageBody))
    {
      v_SdpTemplatePRACK := p_SdpTemplatePRACK;
    }
    v_SdpOffer := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP_Ifpresent(v_PrackRequest.messageBody, v_SdpTemplatePRACK);
    if (isvalue(v_SdpOffer)) {
      f_IMS_MessageHeader_CheckPreconditionIndicationForOffer(v_PrackRequest.msgHeader, PreconditionRequired);     /* check that PRACK contains Require with "preconditions" */
      v_SdpAnswer := f_IMS_BuildSDP_UpdatedAnswer_TX(valueof(v_SdpOffer));
      v_MessageBody200OK := cs_MessageBody_SDP(v_SdpAnswer);
      v_ContentType200OK  := cs_ContentTypeSDP;
      v_PreconditionsRequired200OK := true;
    }

……………………………
  }  


[bookmark: _Toc107254010][bookmark: _Toc112156486]2.2 Change 2
	Function name
	f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7

	Reason for change
	Following implementation of R5-223463, UE may not include SDP content in the PRACK message during the 2nd and 3rd MO call setup sequences, and so content type header may not be included..

	Summary of change
	Add ifpresent to the expected content type header 

Note : Following this change, if the contentType headwer and the SDP body are both missing, the checking of PRACK header contents in f_IMS_MOCallSetup_ReceivePRACK_Send200OK() will succeed but this function will return omit so the sequence will then wait for a subsequent UPDATE request.

	TTCN module
	IMS_CC_CommonFunctions5G.ttcn

	MCC160 Comment
	



Before Change:
	    function f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7(IMS_TestProcedure_Type p_TestProcedure,
                                                             IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,
                                                             charstring p_CalleeContactUri,
                                                             template (omit) SemicolonParam_List p_AditionalContactParams := omit,
                                                             template (value) SDP_Message p_InitialSdpAnswer,
                                                             template (present) SDP_Message p_UpdatedSdpOffer,
                                                             IMS_CallType_Type p_CallType := NORMAL_CALL,
                                                             boolean p_RadioBearersEstablished := false,
                                                             boolean p_EarlyQosConfirmation := true) runs on IMS_PTC return IMS_CC_MO_CallInfo_Type
  { // !!!! PROSE ISSUE: most likely A.19 and A.25 need to allow early QoS confirmation too => p_EarlyQosConfirmation to be removed !!!!
    var IMS_CC_MO_CallInfo_Type v_CallInfo;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);
    var template (value) MessageHeader v_MessageHeader183;
    var template SDP_Message v_SdpOfferPRACK := omit;
    var template ContentType v_ContentTypePRACK := omit;
    var template (omit) SDP_Message v_LatestSdpOffer := omit;

    // step 2                           100 Trying
    f_IMS_MOCallSetup_Send100Trying(p_InviteRequestWithSdp);

    // step 3                           183 Session Progress
    v_MessageHeader183 := f_IMS_InviteResponse_183_MessageHeaderTX(p_InviteRequestWithSdp.Invite, p_CalleeContactUri, p_AditionalContactParams, -, p_CallType);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine183, v_MessageHeader183, cs_MessageBody_SDP(p_InitialSdpAnswer))));

    // step 4 - 5                       PRACK, 200 OK
    if (not p_RadioBearersEstablished) {
      f_IMS_MOCallSetup_ReceivePRACK_Send200OK(p_InviteRequestWithSdp, v_MessageHeader183, -, -, p_CallType);
      f_IMS_CC_TriggerDedicatedBearerActivation(p_TestProcedure);     // to trigger step 13 of 38.508-1 Table 4.9.15.2.2-1
    } else {
      if (p_EarlyQosConfirmation) {
        v_ContentTypePRACK := cr_ContentTypeSDP;                  // @sic R5-223463 sic@ !!!! PROSE ISSUE: not clear whether it should be ifpresent !!!! 
        v_SdpOfferPRACK := p_UpdatedSdpOffer;                     // @sic R5-223463 sic@ !!!! PROSE ISSUE: not clear whether it should be ifpresent !!!!
      } 
      v_LatestSdpOffer := f_IMS_MOCallSetup_ReceivePRACK_Send200OK(p_InviteRequestWithSdp, v_MessageHeader183, v_ContentTypePRACK, v_SdpOfferPRACK, p_CallType);
    }

/**** Skiping code ***********/
  }




After Change:
	  function f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7(IMS_TestProcedure_Type p_TestProcedure,
                                                             IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,
                                                             charstring p_CalleeContactUri,
                                                             template (omit) SemicolonParam_List p_AditionalContactParams := omit,
                                                             template (value) SDP_Message p_InitialSdpAnswer,
                                                             template (present) SDP_Message p_UpdatedSdpOffer,
                                                             IMS_CallType_Type p_CallType := NORMAL_CALL,
                                                             boolean p_RadioBearersEstablished := false,
                                                             boolean p_EarlyQosConfirmation := true) runs on IMS_PTC return IMS_CC_MO_CallInfo_Type
  { // !!!! PROSE ISSUE: most likely A.19 and A.25 need to allow early QoS confirmation too => p_EarlyQosConfirmation to be removed !!!!
    var IMS_CC_MO_CallInfo_Type v_CallInfo;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);
    var template (value) MessageHeader v_MessageHeader183;
    var template SDP_Message v_SdpOfferPRACK := omit;
    var template ContentType v_ContentTypePRACK := omit;
    var template (omit) SDP_Message v_LatestSdpOffer := omit;

    // step 2                           100 Trying
    f_IMS_MOCallSetup_Send100Trying(p_InviteRequestWithSdp);

    // step 3                           183 Session Progress
    v_MessageHeader183 := f_IMS_InviteResponse_183_MessageHeaderTX(p_InviteRequestWithSdp.Invite, p_CalleeContactUri, p_AditionalContactParams, -, p_CallType);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine183, v_MessageHeader183, cs_MessageBody_SDP(p_InitialSdpAnswer))));

    // step 4 - 5                       PRACK, 200 OK
    if (not p_RadioBearersEstablished) {
      f_IMS_MOCallSetup_ReceivePRACK_Send200OK(p_InviteRequestWithSdp, v_MessageHeader183, -, -, p_CallType);
      f_IMS_CC_TriggerDedicatedBearerActivation(p_TestProcedure);     // to trigger step 13 of 38.508-1 Table 4.9.15.2.2-1
    } else {
      if (p_EarlyQosConfirmation) {
        v_ContentTypePRACK := cr_ContentTypeSDP ifpresent; // @sic R5-223463 sic@ !!!! PROSE ISSUE: not clear whether it should be ifpresent !!!!
        v_SdpOfferPRACK := p_UpdatedSdpOffer;                     // @sic R5-223463 sic@ !!!! PROSE ISSUE: not clear whether it should be ifpresent !!!! 
      } 
      v_LatestSdpOffer := f_IMS_MOCallSetup_ReceivePRACK_Send200OK(p_InviteRequestWithSdp, v_MessageHeader183, v_ContentTypePRACK, v_SdpOfferPRACK, p_CallType);
    }

/**** Skiping code ***********/
  }   


[bookmark: _Toc107254012][bookmark: _Toc112156487][bookmark: _Toc107254011]2.3 Change 3
	Function name
	f_IMS_ConferenceCreation_NR5GC_A_19_Steps1_9

	Reason for change
	For conference call TCs, when the calls are merged and the 3rd dialog to the conference factory is created, the function f_IMS_ConferenceCreation_NR5GC_A_19_Steps1_9() sets the parameter p_EarlyQosConfirmation to false when calling f_IMS_MOCallSetup_NR5GC_WithPreconditions_Step2_7(). With this parameter set false the content type header is not allowed in PRACK message, but the UE may still include SDP content in this message.

	Summary of change
	Changed the parameter v_EarlyQosConfirmation value to false

	TTCN module
	IMS_NR5GC_SupplementaryServiceTestcases.ttcn

	MCC160 Comment
	



Before Change:
	  function f_IMS_ConferenceCreation_NR5GC_A_19_Steps1_9(charstring p_ConferenceFactoryUri,
                                                        charstring p_TemporaryConferenceUri,
                                                        charstring p_FinalConferenceUri,
                                                        boolean p_RadioBearersEstablished := false) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var template (value) SemicolonParam_List v_AdditionalContactParams := {cs_GenericParam("isfocus")};
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) SDP_Message v_InitialSdpAnswer;
    var template (present) SDP_Message v_UpdatedSdpOffer;
    var boolean v_EarlyQosConfirmation := false;     // !!!! PROSE ISSUE: most likely early QoS confirmation shall be allowed !!!!

    // Step 1:                  Receive INVITE

/**** Skiping code ***********/

  }  




After Change:
	  function f_IMS_ConferenceCreation_NR5GC_A_19_Steps1_9(charstring p_ConferenceFactoryUri,
                                                        charstring p_TemporaryConferenceUri,
                                                        charstring p_FinalConferenceUri,
                                                        boolean p_RadioBearersEstablished := false) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var template (value) SemicolonParam_List v_AdditionalContactParams := {cs_GenericParam("isfocus")};
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) SDP_Message v_InitialSdpAnswer;
    var template (present) SDP_Message v_UpdatedSdpOffer;
    var boolean v_EarlyQosConfirmation := true;     // !!!! PROSE ISSUE: most likely early QoS confirmation shall be allowed !!!! 

    // Step 1:                  Receive INVITE

/**** Skiping code ***********/

  }    


[bookmark: _Toc112156488]2.4 Change 4
	Function name
	fl_SIP_BuildRefterToUri_Rx()

	Reason for change
	Following implementation of MCC comments for R5s220575, the checking of url parameters in the refer-to header is now too strict - the checking confirms that in the case of 8.36 the “replaces” and “require” headers are present (and no other header) but it now does not allow the header to be present in other cases. In other cases as e.g. 8.34/8.35, the UE may still include the “require” header along with the “replaces” header in the refer-to URI.

	Summary of change
	After adding the mandatory headers as per MCC implementation of R5s220575, add a “superset” operation to allow inclusion of other parameters (similar to the original proposal of R5s220575).

	TTCN module
	

	MCC160 Comment
	



Before Change:
	  function fl_SIP_BuildReferToUri_RX(charstring p_ReferToUriString,
                                     boolean p_ReplacesHeaderIsRequired,
                                     boolean p_RequireHeaderIsRequired) return template (present) SipUrl
  { /* @sic R5s220575: p_RequireHeaderIsRequired sic@ */
    var template (present) SipUrl v_ReferToUri := f_SIP_BuildSipUri_RX(p_ReferToUriString);
    var template (present) UserInfo v_UserInfo;
    var template (present) UriComponents v_UriComponents;
    var template SemicolonParam_List v_Params;

    if (p_ReplacesHeaderIsRequired) {
      if (ischosen(v_ReferToUri.components.tel)) {
        /* convert tel URI to SIP URI acc. to RFC 3261 clause 19.1.6: */
        v_UserInfo := cr_UserInfo(v_ReferToUri.components.tel.subscriber, omit);
        v_Params := superset(cr_GenericParam_Common("user", cr_GenValueToken("phone")));    // @sic R5s160295; NOTE: per default f_SIP_BuildSipUri_RX initialises urlParameters with * sic@
        v_UriComponents := cr_SipUriComponent(v_UserInfo, *, *);
        v_ReferToUri := cr_SipUrl_Common(c_sipScheme, v_UriComponents, v_Params);
      }
      v_ReferToUri.headers[0] := cr_GenericParam_Common("Replaces", cr_GenValueToken(?));
      if (p_RequireHeaderIsRequired) {   /* @sic R5s220575 sic@ */
        v_ReferToUri.headers[1] := cr_GenericParam("Require", "replaces");
      }
    }

    return v_ReferToUri;
  }






After Change:
	 function fl_SIP_BuildReferToUri_RX(charstring p_ReferToUriString,
                                     boolean p_ReplacesHeaderIsRequired,
                                     boolean p_RequireHeaderIsRequired) return template (present) SipUrl
  { /* @sic R5s220575: p_RequireHeaderIsRequired sic@ */
    var template (present) SipUrl v_ReferToUri := f_SIP_BuildSipUri_RX(p_ReferToUriString);
    var template (present) UserInfo v_UserInfo;
    var template (present) UriComponents v_UriComponents;
    var template SemicolonParam_List v_Params;

    if (p_ReplacesHeaderIsRequired) {
      if (ischosen(v_ReferToUri.components.tel)) {
        /* convert tel URI to SIP URI acc. to RFC 3261 clause 19.1.6: */
        v_UserInfo := cr_UserInfo(v_ReferToUri.components.tel.subscriber, omit);
        v_Params := superset(cr_GenericParam_Common("user", cr_GenValueToken("phone")));    // @sic R5s160295; NOTE: per default f_SIP_BuildSipUri_RX initialises urlParameters with * sic@
        v_UriComponents := cr_SipUriComponent(v_UserInfo, *, *);
        v_ReferToUri := cr_SipUrl_Common(c_sipScheme, v_UriComponents, v_Params);
      }
      v_ReferToUri.headers[0] := cr_GenericParam_Common("Replaces", cr_GenValueToken(?));
      if (p_RequireHeaderIsRequired) {   /* @sic R5s220575 sic@ */
        v_ReferToUri.headers[1] := cr_GenericParam("Require", "replaces");
      }
    }
    v_ReferToUri.headers := superset (all from v_ReferToUri.headers);
    return v_ReferToUri;
  }




